14

Hybrid Variational Principles: 1D Application Examples

TABLE OF CONTENTS

§14.1. Introduction

We illustrate the application of the Equilibrium Stress Hybrid (ESH) principle introduced in the precedent Chapter to the derivation of several 1D specific structural elements:

- §9.2: A 2-node, tapered bar element
- §9.3: A 2-node, Bernoulli-Euler-modeled, conventional plane beam element
- §9.4: A 2-node, Bernoulli-Euler-modeled, hinged plane beam element

A 2D example (plate in plane stress) is worked out in the next Chapter.

§14.2. Two-Node Tapered Bar Element

We retake the bar element considered in Chapter 6, which was developed under the Hellinger-Reissner (HR) variational principle. The cross-section tapering law, however, will be different.

The bar member is pictured in Figure 14.1(a). Axes x_1, x_2 and x_3 are relabeled as $\{x, y, y\}$ for convenience. The $x_1 \equiv x$ axis is placed along the longitudinal direction whereas axes y and z lie on the rightmost cross section as shown. The material is isotropic with elastic modulus E. Body forces are ignored. The bar length is L. The bar cross section is a solid square of side h, which varies *linearly* from h_1 at end $x = 0$ to h_2 at end $x = L$. The ratio h_2/h_1 is called β.



FIGURE 14.1. Two-node tapered bar element by ESH principle: (a) shows the bar as a 3D object, and (b) as a FEM model.

§14.2.1. Bar Element Description

The bar member is modeled by a single 2-node element, as illustrated in Figure 14.1(b). Cross section positions are conveniently located in terms of the isoparametric coordinate ξ that varies from $\xi = -1$ at end $x = 0$ (node 1) through $\xi = +1$ at $x = L$ (node 2). The variation of h along the bar can be expressed using isoparametric interpolation as

$$h(\xi) = \tfrac{1}{2}(1-\xi)\,h_1 + \tfrac{1}{2}(1+\xi)\,h_2 = \frac{h_1+h_2}{2} + \frac{h_2-h_1}{2}\xi = \frac{h_1}{2}\Big(1+\beta+(1-\beta)\xi\Big). \quad (14.1)$$

The cross section area varies as $A(\xi) = h(\xi)^2$; in particular $A_1 = h_1^2$ and $A_2 = h_2^2 = \beta^2 A_1$. The mean sectional area is given by

$$A_m = \frac{1}{L}\int_0^L A(x)\,dx = \frac{L/2}{L}\int_{-1}^{+1} A(\xi)\,d\xi = \tfrac{1}{3}A_1(1+\beta+\beta^2). \tag{14.2}$$

The degrees of freedom of the element are the end displacements u_1 and u_1 and the parameters that define the axial force N. The nodal forces associated with u_1 and u_2 are f_1 and f_2, respectively, which are positive if directed along $+x$.

§14.2.2. The ESH Bar Functional

The reduction of the general Equilibrium Stress Hybrid (ESH) functional to the special case of the bar of Figure 14.1(a) can be done following the same specialization steps as those discussed for the HR principle in §11.3. As in that case the varied fields are the axial force N and the interface displacements d at the bar ends (element nodes). The end result is

$$\boxed{\begin{aligned}\Pi_{ESH}[u,N] &= \int_0^L \left(-\frac{N^2}{2EA}\right) dx + N\,d|_0^L + f_1\,d_1 - f_2\,d_2 \\ &= \int_0^L \left(-\frac{N^2}{2EA}\right) dx + N_1\,u_1 + N_2\,u_2 - f_1\,u_1 - f_2\,u_2.\end{aligned}} \tag{14.3}$$

For the second expression we have identified $d|_0 = d_1$ and $d|_L = d_2$ with the node displacements $-u_1$ and u_2, respectively. (Notice that $d_1 = -u_1$ because by convention d_1 is positive along the exterior normal at $x = 0$, which is oriented along $-x$.)

§14.2.3. Bar Element Formulation

To construct the element e depicted in Figure 14.1(b), from the functional (14.3), we make assumptions on the two masters: axial force and interface displacements:

1. The axial force N must satisfy strongly the equlibrium equation $N' + q = 0$. Since the body force q vanishes by assumption, it is sufficient to assume that $N = N^e$ is *constant* over the element.
2. Interface displaments can be directly identified with nodal displacements. Hence $u_1 = u_1^e = -d_1$ and $u_2 = u_2^e = d_2$. The conjugate forces are $f_1 = f_1^e$ and $f_2 = f_2^e$. Nothing more needs to be done; in particular the internal displacement $u^e(x)$ for $0 < x < L$ is of no concern.

From the axial force assumption: $N_1 = N_2 = N^e$, and the functional (14.3) reduces to the algebraic function

$$\boxed{\Pi^e_{ESH}(u_1^e, u_2^e, N^e) = \int_0^L -\frac{(N^e)^2}{2EA}\,dx - N^e(u_1^e + u_2^e) - f_1\,u_1^e - f_2\,u_2^e.} \tag{14.4}$$

The only complication in evaluating (14.4) is that A, which appears in the denominator of the integral term, varies along the bar. Using *Mathematica* it is found that this integral is quite simple:

$$\int_0^L \frac{(N^e)^2}{2EA}\,dx = \frac{L}{2}\int_{-1}^{+1}\frac{(N^e)^2}{2EA(\xi)}\,d\xi = \frac{(N^e)^2 L}{2EA_1\beta} = \frac{1+\beta+\beta^2}{3\beta}\frac{(N^e)^2 L}{2E\,A_m} \stackrel{\text{def}}{=} \gamma\frac{(N^e)^2 L}{2E\,A_m}. \tag{14.5}$$

in which $\gamma = (1 + \beta + \beta^2)/(3\beta)$ is a dimensionless function of the ratio $\beta = h_2/h_1$, and A_m is the mean section area given by (14.2). Substitution into (14.4) yields

$$\Pi^e_{ESH}[u^e_1, u^e_2, N^e] = -\frac{\gamma\,(N^e)^2 L}{2E\,A_m} + N^e(u^e_2 - u^e_1) - f^e_1\,u^e_1 - f^e_2\,u^e_2. \tag{14.6}$$

The finite element equations are obtained from the stationary conditions

$$\frac{\partial \Pi^e_{ESH}}{\partial N^e} = \frac{\partial \Pi^e_{ESH}}{\partial u^e_1} = \frac{\partial \Pi^e_{ESH}}{\partial u^e_2} = 0, \tag{14.7}$$

which explicitly read $-\gamma\,N^e\,L/(E\,A_m) + u^e_2 - u^e_1 = 0$, $-N^e + f^e_1 = 0$, and $N^e_2 + f^e_2 = 0$. The last two are nodal equilibrium equations (to verify this, draw the FBD of each end node), whereas the first one expresses axial kinematic compatibility: the bar elongation $u^e_2 - u^e_1$ is $\gamma\,N^e\,L/(E\,A_m)$. Writing these three expressions in matrix form yields the finite element equations

$$\begin{bmatrix} -\dfrac{\gamma L}{E\,A_m} & -1 & 1 \\ -1 & 0 & 0 \\ 1 & 0 & 0 \end{bmatrix} \begin{bmatrix} N^e \\ u^e_1 \\ u^e_2 \end{bmatrix} = \begin{bmatrix} 0 \\ f^e_1 \\ f^e_2 \end{bmatrix}, \quad \text{in which} \quad \gamma = \frac{1+\beta+\beta^2}{3\beta}. \tag{14.8}$$

These can be presented into the partitioned matrix form introduced in Chapter 11 (the HR principle), which is standard for HR and ESH formulations:

$$\begin{bmatrix} -\mathbf{F} & \mathbf{G} \\ \mathbf{G}^T & \mathbf{0} \end{bmatrix} = \begin{bmatrix} \mathbf{0} \\ \mathbf{f}^e \end{bmatrix}, \tag{14.9}$$

Here $\mathbf{F}$ is a 1×1 *flexibility matrix* whereas $\mathbf{G}$ is a 1×2 *connection matrix*.

Because the axial-force degree of freedom N^e is not continuous across elements (recall that C^{-1} continuity for stress variables is allowed in the ESH principle because their variational index is zero), it may be eliminated or "condensed out" at the element level. The static condensation process studied in Chapter 10 of IFEM yields

$$\frac{E\,A_m}{\gamma L}\begin{bmatrix} 1 & -1 \\ -1 & 1 \end{bmatrix} \begin{bmatrix} u^e_1 \\ u^e_2 \end{bmatrix} = \begin{bmatrix} f^e_1 \\ f^e_2 \end{bmatrix}, \tag{14.10}$$

or

$$\boxed{\mathbf{K}^e\,\mathbf{u}^e = \mathbf{f}^e.} \tag{14.11}$$

These are the *element stiffness equations*, obtained here through the ESH principle. Had these equations been derived through the TPE principle, one would have obtained a similar expression except that $\gamma = 1$ for any end-area ratio. Thus if the element is prismatic ($A_1 = A_2 = A_m$, or $\beta = 1$) the ESH and TPE functionals lead to the same element stiffness equations.



FIGURE 14.2. Ratio of end displacements: u_2^{TPE}/u_2^{ESH} as function of $\beta = h_2/h_1$ for the one-element discretization of Figure 14.1(b), with u_2^{ESH} providing the exact elasticity solution.

Table 14.1 Results for one-element analysis of fixed-free tapered bar

End dimensions ratio	u_2 from ESH	u_2 from TPE	Exact u_2
$\beta = h_2/h_1 = 1$	$PL/(EA_m)$	$PL/(EA_m)$	$PL/(EA_m)$
$\beta = h_2/h_1 = 1/2$	$1.167PL/(EA_m)$	$PL/(EA_m)$	$1.167PL/(EA_m)$
$\beta = h_2/h_1 = 1/5$	$2.067PL/(EA_m)$	$PL/(EA_m)$	$2.067PL/(EA_m)$

Example 14.1. To give a simple numerical example, suppose that the bar of Figure 14.1(a) is fixed at end 1 whereas end 2 is under a given axial force P. Results for sample end area ratios are given in Table 14.1. It can be seen that the ESH formulation yields the exact displacement solution for all tapering ratios. The ratio of the two results is plotted in Figure 14.2 as a function of β.

Also note that the discrepancy of the one-element TPE solution from the exact one grows as the cross section dimension taper ratio β strays from one. If $\beta < 1$, TPE elements *underestimate* the actual deflections, and are therefore on the stiff side. For instance, if $\beta = 5$ the TPE element is in error by over 50%. To improve the TPE results one would need either to divide the bar into more 2-node elements, or to use higher order displacement elements, such as the 3-node quadratic model.

§14.3. Two-Node, Plane, B-E Beam Element

This Section takes a second look at the beam element derived in Chapter 12 of IFEM. The formulation used over there relies on displacement assumptions and the Total Potential Energy (TPE) functional. It is redone here using an Equilibrium Stress Hybrid (ESH) variational principle. The procedure delivers good and bad news. Good news: simpler shape functions (linear instead of cubic) can be used; consequently element integrals are simplified. Bad news: the same element stiffness equations result after condensation. So the main gain is practical exposure to a new method. The hinged-beam case worked out in the next Section is more favorable to the ESH approach.

We deal with the plane, prismatic, beam member shown in Figure 14.3(a). Its mechanical behavior is idealized by the Bernoulli-Euler (BE) model, also called engineering beam theory. This model only



FIGURE 14.3. Two-node, plane, prismatic, Bernoulli-Euler beam element to be developed with the equilibrium-stress hybrid (ESH) formulation.



FIGURE 14.4. Degrees of freedom and conjugate actions for hinged beam element of Figure 14.4.

accounts for internal energy due to bending, ignoring contribution of the transverse shear. Cartesian axes $\{x, y, z\}$ are chosen as shown in that Figure. As usual, a prime will denote differentiation with respect to x, so $(.)' = d(.)/dx$. The elastic modulus E and the second moment of inertia $I_{zz} = \int_A y^2\, dA$ are constant along x. The member is acted upon by a distributed lateral load $q(x)$, expressed as force per unit length, which may represent a distributed dead or live load.

The lateral displacement, positive along $+y$, is denoted by $v(x)$. The cross section rotation about its neutral axis is, according to the BE model, $\theta(x) = dv(x)/dx = v'(x)$, positive CCW. The internal member forces include the bending moment $M(x)$ and the transverse shear force $V(x) = dM(x)/dx = M'(x)$, with positive sign conventions illustrated in Figure 14.3(b).

§14.3.1. The ESH Principle

The ESH variational principle for the Bernoulli-Euler (BE) plane beam model is most readily developed by starting from the classical Hellinger-Reissner (HR) principle introduced in Chapter 5. For an *individual* element such as that shown in Figure 14.3(c,d,e)

$$\boxed{\Pi^e_{HR} = \int_0^{L^e} \left[-\frac{M^2}{2E\,I_{zz}} + M\,v'' - q v\right] dx - \left(m^e_1\,\theta^e_1 + m^e_2\,\theta^e_2 + f^e_1\,v^e_1 + f^e_2\,v^e_2\right),} \tag{14.12}$$

in which the master fields are $M = M^e(x)$ and $v = v^e(x)$. (The e superscripts in M and v are omitted in the integral term of (14.12) to reduce clutter.) The end forces and moments f^e_1, f^e_2, m^e_1 and m^e_2, defined in Figure 14.4(b) as work-conjugate actions to the element DOF, are viewed as prescribed.[1] If the bending moment $M(x)$ satisfies strongly the equilibrium (balance) equation so that $M'' = d^2M^e(x)/dx^2 = q$, the combination of $M\,v''$ and $q\,v$ terms may be converted, upon double integration by parts of the Mv'' term, to a function of end values:

$$\begin{aligned}\int_0^{L^e} \left(M\,v'' - q\,v\right) dx &= \int_0^{L^e} -\left(M'\,v' + q\,v\right) dx + M\,v'\big|_{x=0}^{x=L^e} \\ &= \int_0^{L^e} \left(M'' \cancelto{0}{-q}\right) v\,dx + M\,v'\big|_{x=0}^{x=L^e} - M'\,v\big|_{x=0}^{x=L^e} = M\,v'\big|_{x=0}^{x=L^e} - M'\,v\big|_{x=0}^{x=L^e}.\end{aligned} \tag{14.13}$$

We thus obtain the ESH functional

$$\boxed{\begin{aligned}\Pi^e_{ESH} &= U^e_{ESH} - W^e_{ESH}, \\ U^e_{ESH} &= -\int_0^{L^e} \frac{M^2\,dx}{2E\,I_{zz}} + M\,v'\big|_{x=0}^{x=L^e} - M'\,v\big|_{x=0}^{x=L^e} \\ &= -\int_0^{L^e} \frac{M^2\,dx}{2E\,I_{zz}} - M^e_1\,\theta^e_1 + M^e_2\,\theta^e_2 + V^e_1\,v^e_1 + V^e_2\,v^e_2, \\ W^e_{ESH} &= m^e_1\,\theta_1 + m^e_2\,\theta^e_2 + f^e_1\,v^e_1 + f^e_2\,v^e_2.\end{aligned}} \tag{14.14}$$

Here the second form of U^e_{ESH} results from the substitutions at the end nodes[2]

[1] Do not confuse m^e_1 with $M^e_1 = M(0)$, say. The latter is an internal field evaluated at node 1, not a prescribed value.

[2] V^e_2 is conventionally chosen as $-M'(L^e)$ so it is work conjugate to v^e_2, which is positive along $+y$. If one picks $V^e_2 = +M'(L^e)$ instead, term $V^e_2 v^e_2$ in (14.14) should have a minus sign.

$$
\begin{aligned}
&M_1^e = M(0), \quad M_2^e = M(L^e), \quad V_1^e = M'(0), \quad V_2^e = -M'(L^e),\\
&v_1^e = v(0), \quad v_2^e = v(L^e), \quad \theta_1^e = v'(0), \quad \theta_2^e = v'(L^e).
\end{aligned}
\tag{14.15}
$$

The interior displacements $v(x)$ have disappeared from Π^e_{ESH}. This is the hybrid principle used for the beam element derivation that follows.

§14.3.2. ESH Beam Element Derivation

The two master fields to be varied in Π^e_{ESH} are:

1. The end displacements and rotations. These can be identified with the nodal freedoms $v_1^e = v(0)$, $v_2^e = v(L^e)$, $\theta_1^e = v'(0)$ and $\theta_2^e = v'(L^e)$ pictured in Figure 14.4(a).
2. The bending moment $M(x)$. To simplify the moment assumption, we take $q(x) \equiv 0$ over the element, whence lateral forces can only be applied at end nodes. The moment variation can then be *linear* in x because if $q = 0$ the BE-beam equilibrium equation $M'' = q = 0$ is verified strongly. We take

$$
M^e(x) \stackrel{\text{def}}{=} M^e(\xi) = M_{ref}\big(\beta_1^e N_1 + \beta_2^e N_2\big), \quad N_1 = \tfrac{1}{2}(1-\xi), \quad N_2 = \tfrac{1}{2}(1+\xi). \tag{14.16}
$$

   Here $\xi = 2x/L^e - 1$ is the natural (isoparametric) coordinate defined in Figure 14.3(f), β_1^e and β_2^e are *stress parameters*, N_1 and N_2 are the well known linear shape functions, and M_{ref} is a reference moment introduced only to make β_1^e and β_2^e dimensionless.[3]

The transverse shear associated with (14.16) is

$$
V^e(x) = \frac{dM^e(x)}{dx} = \frac{dM^e(\xi)}{d\xi}\frac{d\xi}{dx} = \tfrac{1}{2}(\beta_2^e - \beta_1^e)\,M_{ref}\,\frac{2}{L^e} = \frac{(\beta_2^e - \beta_1^e)\,M_{ref}}{L^e}, \tag{14.17}
$$

which is *constant* over the element. Evaluation at the end nodes gives $V_1^e = (\beta_2^e - \beta_1^e)\,M_{ref}/L^e$ and $V_2^e = -V_1^e$ (as regard the sign of the latter see footnote on previous page.) Inserting these assumptions into Π^e_{ESH} yields

$$
\begin{aligned}
\Pi^e_{ESH} = &-\frac{(L^e)^2(M_{ref})^2}{6E\,I_{zz}}(\beta_1^e\beta_1^e + \beta_1^e\beta_2^e + \beta_2^e\beta_2^e) + M_{ref}\beta_1^e(-v_1^e/L^e - \theta_1^e + v_2^e/L^e)\\
&+ M_{ref}\beta_2^e(v_1^e/L^e - v_2^e/L^e + \theta_2^e) - f_1^e\,v_1^e - m_1^e\,\theta_1^e - f_2^e\,v_2^e - m_2^e\,\theta_2^e.
\end{aligned}
\tag{14.18}
$$

This function (no longer a functional) must be stationary with respect to variations in the six degrees of freedom. Consequently

$$
\frac{\partial\Pi^e_{ESH}}{\partial\beta_1^e} = \frac{\partial\Pi^e_{ESH}}{\partial\beta_2^e} = \frac{\partial\Pi^e_{ESH}}{\partial v_1^e} = \frac{\partial\Pi^e_{ESH}}{\partial\theta_1^e} = \frac{\partial\Pi^e_{ESH}}{\partial v_2^e} = \frac{\partial\Pi^e_{ESH}}{\partial\theta_2^e} = 0. \tag{14.19}
$$

[3] This M_{ref} will disappear upon condensation of the stress parameters.

```
ClearAll[EI,Le,ξ,Mref,v1e,θ1e,v2e,θ2e,m1e,m2e,f1e,f2e,βe1,βe2];
Mxe=Mref*(βe1*(1-ξ)/2+βe2*(1+ξ)/2);
Uce=Simplify[Integrate[-Mxe^2/(2*EI)*(Le/2),{ξ,-1,1}]];
M1e=Mxe/.{ξ->-1}; M2e=Mxe/.{ξ->1}; V1e=(M2e-M1e)/Le; V2e=-V1e;
Ube=M2e*θ2e-M1e*θ1e+V1e*v1e+V2e*v2e;
We=m1e*θ1e+m2e*θ2e+f1e*v1e+f2e*v2e;
Πe=Simplify[Uce+Ube-We];
uβe={βe1,βe2,v1e,θ1e,v2e,θ2e};
Keβ=StiffnessMatrixAsHessian[Πe,uβe];
Print["Keβ="," Mref*",Simplify[Keβ/Mref]//MatrixForm];
Ke=CondenseMatrixFreedoms[Keβ,{1,2}];
Print["Ke=","EI/(Le^3)*",Simplify[Ke*Le^3/EI]//MatrixForm];
fe=ForceVectorAsGradient[Πe,uβe]; Print["fe=",fe];
KeTPE={{12,6*Le,-12,6*Le},{6*Le,4*Le^2,-6*Le,2*Le^2},
     {-12,-6*Le,12,-6*Le},{6*Le,2*Le^2,-6*Le,4*Le^2}}*EI/Le^3;
Print["check vs IFEM:",Simplify[Ke-KeTPE]//MatrixForm];
```

FIGURE 14.5. Script for derivation of the two-node conventional plane beam element using the ESH variational principle. Utility modules `StiffMatrixAsHessian`, `ForceVectorAsGradient`, `CondenseMatrixFreedom` and `CondenseMatrixFreedoms` are listed in Figure 14.6.

```
StiffnessMatrixAsHessian[U_,ue_]:=Module[{i,j,ndof=Length[ue],
      re,Ke}, re=Table[D[U,ue[[i]]],{i,1,ndof}];
   Ke=Table[Table[D[re[[i]],ue[[j]]],{i,1,ndof}],{j,1,ndof}];
   Ke=Simplify[Ke]; ClearAll[re];
   Return[Ke]];

ForceVectorAsGradient[U_,ue_]:=Module[{i,j,ndof=Length[ue],fe},
   fe=-Table[D[U,ue[[i]]],{i,1,ndof}];
   For [i=1,i<=ndof,i++, fe=fe/.ue[[i]]->0];
   fe=Simplify[fe];
   Return[fe]];

CondenseMatrixFreedom[K_,k_]:=Module[{c,pivot,Kc,ii,jj,
   n=Length[K],modname="CondenseMatrixFreedom: "},
   If [n<=0,Return[K]]; If [k<=0 || k>n, Return[K]];
   Kc=Table[0,{n-1},{n-1}]; pivot=K[[k,k]];
   If [pivot==0, Print[modname,
         "zero pivot row ",k]; Return[K]]; ii=0;
   For [i=1,i<=n,i++,  If [i==k, Continue[]]; ii++;
        c=K[[i,k]]/pivot; jj=0;
        For [j=1,j<=n,j++, If [j==k,Continue[]]; jj++;
             Kc[[ii,jj]]=K[[i,j]]-c*K[[k,j]]
            ];
       ]; Kc=Simplify[Kc];
       Return[Kc]
   ];

CondenseMatrixFreedoms[K_,ind_]:=Module[{Kc,Kcc,i,k,m,inds},
   inds=Sort[ind,Greater]; m=Length[inds]; Kc=K;
   For [i=1,i<=m,i++, k=inds[[i]];
        Kcc=CondenseMatrixFreedom[Kc,k];
        Kc=Kcc];
   ClearAll[Kcc]; Return[Kc]];
```

FIGURE 14.6. Utility modules for symbolic derivation of finite element equations.

Equations (14.19) may be expressed in matrix form as

$$M_{ref}\begin{bmatrix} -\frac{M_{ref}L^e}{3EI_{zz}} & -\frac{M_{ref}L^e}{6EI_{zz}} & -1/L^e & -1 & 1/L^e & 0 \\ -\frac{M_{ref}L^e}{6EI_{zz}} & -\frac{M_{ref}L^e}{3EI_{zz}} & 1/L^e & 0 & -1/L^e & 1 \\ -1/L^e & 1/L^e & 0 & 0 & 0 & 0 \\ -1 & 0 & 0 & 0 & 0 & 0 \\ 1/L^e & -1/L^e & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \end{bmatrix}\begin{bmatrix} \beta_1^e \\ \beta_2^e \\ v_1^e \\ \theta_1^e \\ v_2^e \\ \theta_2^e \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ f_1^e \\ m_1^e \\ f_2^e \\ m_2^e \end{bmatrix}. \tag{14.20}$$

The configuration (14.20) befits the partitioned matrix form often encountered with mixed and hybrid formulations:

$$\begin{bmatrix} -\mathbf{F} & \mathbf{G} \\ \mathbf{G}^T & \mathbf{0} \end{bmatrix}\begin{bmatrix} \boldsymbol{\beta}^e \\ \mathbf{u}^e \end{bmatrix} = \begin{bmatrix} \mathbf{0} \\ \mathbf{f}^e \end{bmatrix}, \tag{14.21}$$

in which

$$\mathbf{F} = \frac{M_{ref}^2\, L^e}{6EI_{zz}}\begin{bmatrix} 2 & 1 \\ 1 & 2 \end{bmatrix}, \quad \mathbf{G} = M_{ref}\begin{bmatrix} -1/L^e & -1 & 1/L^e & 0 \\ 1/L^e & 0 & -1/L^e & 1 \end{bmatrix}, \tag{14.22}$$
$$\boldsymbol{\beta}^e = [\,\beta_1^e \quad \beta_2^e\,]^T, \quad \mathbf{u}^e = [\,v_1^e \quad \theta_1^e \quad v_2^e \quad \theta_2^e\,]^T.$$

Here **F** is a *flexibility matrix* whereas **G** is a *connection matrix*. Since the variational index of the bending moment function in Π^e_{ESH} is 0, only C^{-1} interelement continuity is required for this master field. Therefore the stress parameters and β_1^e and β_2^e may be statically condensed at the element level, which gives the element stiffness equations

$$\mathbf{K}^e\,\mathbf{u}^e = \mathbf{f}^e, \qquad \mathbf{K} = \mathbf{G}^T\mathbf{F}^{-1}\mathbf{G}. \tag{14.23}$$

In detail

$$\boxed{\frac{EI_{zz}}{(L^e)^3}\begin{bmatrix} 12 & 6L^e & -12 & 6L^e \\ 6L^e & 4(L^e)^2 & -6L^e & 2(L^e)^2 \\ -12 & -6L^e & 12 & -6L^e \\ 6L^e & 2(L^e)^2 & -6L^e & 4(L^e)^2 \end{bmatrix}\begin{bmatrix} v_1^e \\ \theta_1^e \\ v_2^e \\ \theta_2^e \end{bmatrix} = \begin{bmatrix} f_1^e \\ m_1^e \\ v_2^e \\ m_2^e \end{bmatrix}.} \tag{14.24}$$

Observe that M_{ref} is gone. The foregoing computations were carried out using the *Mathematica* script listed in Figure 14.5. In turn this code makes use of the utility modules for symbolic element derivation listed in Figure 14.6.

Comparing (14.24) with the TPE formulation worked out in Chapter 12 of IFEM it can be seen that exactly the same stiffness equations are obtained. In the **Notes and References** at the end of this Chapter[4] it is noted that the coalescence is to be expected in this particular case on account of Fraeijs de Veubeke's *limitation principle*. The conclusion would be different, however, should element properties vary along its length, as in the next example.

[4] to be eventually appended

§14.4. Two-Node, Hinged, Plane B-E Beam Element

This element differs from that treated in the previous Section in the presence of a *rotational hinge* within the element span. At the hinge section the bending moment vanishes. As a result the lateral displacement expression gets more complicated whereas the moment assumption is simpler. These changes favor the ESH approach.

Figure 14.7(a) illustrates a plane, prismatic, *hinged* beam member. Its mechanical behavior away from the hinged section is idealized by the Bernoulli-Euler (BE) model. Cartesian axes $\{x, y, z\}$ are chosen as shown in that Figure. Again a prime will denote differentiation with respect to x, so $(.)' = d(.)/dx$. Except for the hinged section, the elastic modulus E and the second moment of inertia $I_{zz} = \int_A y^2\, dA$ are constant along x. The member is acted upon by a distributed lateral load $q(x)$, expressed as force per unit length. The lateral displacement, positive along $+y$, is denoted by $v(x)$. The cross section rotation about its neutral axis is, according to the BE model, $\theta(x) = dv(x)/dx = v'(x)$, positive CCW. The internal forces include the bending moment $M(x)$ and the transverse shear force $V(x) = dM(x)/dx = M'(x)$, with positive sign conventions illustrated in Figure 14.7(b).

Upon breaking up the member into finite elements, one encounters two element types: hinged and non-hinged. Non-hinged, 2-node beam elements may be derived either with the TPE principle as covered in Chapter 12 of IFEM, or through the ESH approach as done in §14.3. For the hinged element, the ESH approach leads to the element stiffness equations faster than going through TPE, because it bypasses the error-prone use of complicated (piecewise cubic) shape functions that must allow a "kink" (slope discontinuity) at the hinge section.

Notation for the hinged element is introduced pictorially in Figure 14.7(c,d,e) and in Figure 14.8.

The ESH functional Π_{ESH} is the same as that derived in §14.3.1. Consequently (14.14) can be reused for the hinged element.

§14.4.1. Element Assumptions

As in §14.3.2, the two master fields to be varied in Π^e_{ESH} are:

1. The bending moment $M^e(x)$ over the element.
2. The interface DOF: end displacements and rotations. These can be identified with the nodal freedoms $v^e_1 = v(0)$, $v^e_2 = v(L^e)$, $\theta^e_1 = v'(0)$ and $\theta^e_2 = v'(L^e)$ pictured in Figure 14.8(a).

To simplify the moment assumption, we again take $q(x) \equiv 0$ over the element; thus lateral forces are only applied at end nodes. Consequently the moment variation can be *linear* in x because if so the equilibrium equation $M'' = q = 0$ is verified. Since $M^e(x)$ must vanish at the hinge, we take

$$M^e(x) \stackrel{\text{def}}{=} M^e(\xi) = \beta^e\, M_{ref}\, (\xi - \xi_H) \tag{14.25}$$

Here $\xi = 2x/L^e - 1$ is the natural (isoparametric) coordinate defined in Figure 14.7(f), $-1 \le \xi_H \le 1$ is the natural coordinate of the hinge, β^e is a *stress parameter* and M_{ref} is a reference moment introduced only to make β^e dimensionless. (This M_{ref} will disappear upon moment condensation.) The interpretation of β^e as a *scaled slope* of $M^e(x)$ is noted in Figure 14.7(e).

The transverse shear associated with (14.25) is

$$V^e(x) = \frac{dM^e(x)}{dx} = \frac{dM^e(\xi)}{d\xi}\frac{d\xi}{dx} = \beta^e\, M_{ref}\, \frac{2}{L^e}. \tag{14.26}$$



FIGURE 14.7. Two-node, plane, prismatic, hinged, Bernoulli-Euler-modeled beam element to be formulated by a ESH variational principle.



FIGURE 14.8. Degrees of freedoms and conjugate actions for hinged beam element of Figure 14.7.

This is *constant* over the element. End node evaluation gives $V_1^e = 2\beta_2^e\, M_{ref}/L^e$ and $V_2^e = -V_1^e$. For the interface displacements we simply pick the nodal DOF as stated above. Note that trying to express $v^e(x)$ within the element would be messy because it consists of two cubic polynomial patches with a "kink" at the hinge, as pictured in Figure 14.7(f). We should not be concerned about this complication, however, since the inner $v^e(x)$ has disappeared in the ESH functional. This feature makes the derivation much simpler.

§14.4.2. Element Derivation

Inserting the above moment, shear and interface-displacement assumptions into Π^e_{ESH}, and integrating the complementary energy term $\left(M^e(x)\right)^2/(2E\,I_{zz})$ over the element, the funtional reduces to an ordinary algebraic function of the degrees of freedom:

$$\begin{aligned}\Pi^e_{ESH} = &-\frac{L^e\, M^2_{ref}\,(\beta^e)^2(1{+}3\xi_H^2)}{6E\,I_{zz}} + M_{ref}\,\beta^e\left(\frac{2v_1^e}{L^e} + \theta_1^e(1{+}\xi_H) - \frac{2v_2^e}{L^e} + \theta_2^e(1{-}\xi_H)\right)\\ &-\, f_1^e\, v_1^e - m_1^e\,\theta_1^e - \, f_2^e\, v_2^e - m_2^e\,\theta_2^e.\end{aligned} \tag{14.27}$$

This function must be stationary with respect to variations in the five degrees of freedom:

$$\frac{\partial\Pi^e_{ESH}}{\partial\beta^e} = \frac{\partial\Pi^e_{ESH}}{\partial v_1^e} = \frac{\partial\Pi^e_{ESH}}{\partial\theta_1^e} = \frac{\partial\Pi^e_{ESH}}{\partial v_2^e} = \frac{\partial\Pi^e_{ESH}}{\partial\theta_2^e} = 0. \tag{14.28}$$

These five linear equations may be arranged in matrix form:

$$M_{ref}\begin{bmatrix} -\frac{M_{ref}L^e(1{+}3\xi_H^2)}{3E\,I_{zz}} & \frac{2}{L^e} & 1+\xi_H & -\frac{2}{L^e} & 1-\xi_H \\ \frac{2}{L^e} & 0 & 0 & 0 & 0 \\ 1+\xi_H & 0 & 0 & 0 & 0 \\ -\frac{2}{L^e} & 0 & 0 & 0 & 0 \\ 1-\xi_H & 0 & 0 & 0 & 0 \end{bmatrix}\begin{bmatrix}\beta^e\\ v_1^e\\ \theta_1^e\\ v_2^e\\ \theta_2^e\end{bmatrix} = \begin{bmatrix}0\\ f_1^e\\ m_1^e\\ f_2^e\\ m_2^e\end{bmatrix}. \tag{14.29}$$

The configuration (14.29) befits the partitioned matrix form often encountered with mixed and hybrid formulations:

$$\begin{bmatrix}-\mathbf{F} & \mathbf{G}\\ \mathbf{G}^T & \mathbf{0}\end{bmatrix}\begin{bmatrix}\boldsymbol{\beta}^e\\ \mathbf{u}^e\end{bmatrix} = \begin{bmatrix}\mathbf{0}\\ \mathbf{f}^e\end{bmatrix}, \tag{14.30}$$

in which

$$\begin{aligned}&\mathbf{F} = \left[\,\frac{M^2_{ref}\,L^e\,(1{+}3\xi_H^2)}{3E\,I_{zz}}\,\right],\quad \mathbf{G} = M_{ref}\left[\,\frac{2}{L^e}\quad 1+\xi_H\quad -\frac{2}{L^e}\quad 1-\xi_H\,\right],\\ &\boldsymbol{\beta}^e = [\,\beta^e\,],\quad \mathbf{u}^e = [\,v_1^e\quad \theta_1^e\quad v_2^e\quad \theta_2^e\,]^T.\end{aligned} \tag{14.31}$$

Here $\mathbf{F}$ is a *flexibility matrix* whereas $\mathbf{G}$ is a *connection matrix*. Since the variational index of the bending moment function in Π^e_{ESH} is 0, only C^{-1} interelement continuity is required for this master field. Therefore the stress parameter β^e may be statically condensed at the element level, which gives the element stiffness equations

$$\mathbf{K}^e\,\mathbf{u}^e = \mathbf{f}^e, \qquad \mathbf{K} = \mathbf{G}^T\mathbf{F}^{-1}\mathbf{G}. \tag{14.32}$$

```
ClearAll[EI,Le,ξ,ξH];
ClearAll[Mref,v1e,θ1e,v2e,θ2e,m1e,m2e,f1e,f2e,βe];
Mxe=βe*Mref*(ξ-ξH);  J=Le/2;
Uce=Simplify[Integrate[-Mxe^2/(2*EI)*J,{ξ,-1,1}]];
M1e=Mxe/.{ξ->-1}; M2e=Mxe/.{ξ->1}; V1e=(M2e-M1e)/Le; V2e=-V1e;
Ube=Simplify[M2e*θ2e-M1e*θ1e+V1e*v1e+V2e*v2e];
We=Simplify[m1e*θ1e+m2e*θ2e+f1e*v1e+f2e*v2e];
Πe=Simplify[Uce+Ube-We]; Print["Πe=",Πe//InputForm];
uβe={βe,v1e,θ1e,v2e,θ2e}; Keβ=StiffnessMatrixAsHessian[Πe,uβe];
Print["Keβ=","  Mref*",Simplify[Keβ/Mref]//MatrixForm];
Ke=CondenseMatrixFreedoms[Keβ,{1}];
Print["Ke=","3*EI/(Le*(1+3*ξH^2))*",
       Simplify[Ke*(1+3*ξH^2)*Le/(3*EI)]//MatrixForm];
Print["Eigs of Ke=",Simplify[Eigenvalues[Ke]]];
fe=ForceVectorAsGradient[Πe,uβe]; Print["fe=",fe];
Ke0=Simplify[Ke/.ξH-> 0]; Print["Ke0=",Ke0//MatrixForm];
Ke0TPE={{4,2*Le,-4,2*Le},{2*Le,Le^2,-2*Le,Le^2},
 {-4,-2*Le,4,-2*Le},{2*Le,Le^2,-2*Le,Le^2}}*(3*EI/Le^3);
Print["check vs TPE derived element:",Simplify[Ke0-Ke0TPE]];
```

FIGURE 14.9. *Mathematica* script for derivation of the two-node hinged beam element. Utility modules `StiffMatrixAsHessian`, `ForceVectorAsGradient`, `CondenseMatrixFreedom` and `CondenseMatrixFreedoms` are listed in Figure 14.6.

Upon condensation, the element stiffness equations in detail are

$$\frac{3EI_{zz}}{L^e(1+3\xi_H^2)}\begin{bmatrix} \frac{4}{(L^e)^2} & \frac{2(1+\xi_H)}{L^e} & -\frac{4}{(L^e)^2} & \frac{2(1-\xi_H)}{L^e} \\ \frac{2(1+\xi_H)}{L^e} & (1+\xi_H)^2 & -\frac{2(1+\xi_H)}{L^e} & 1-\xi_H^2 \\ -\frac{4}{(L^e)^2} & -\frac{2(1+\xi_H)}{L^e} & \frac{4}{(L^e)^2} & -\frac{2(1-\xi_H)}{L^e} \\ \frac{2(1-\xi_H)}{L^e} & 1-\xi_H^2 & -\frac{2(1-\xi_H)}{L^e} & (1-\xi_H)^2 \end{bmatrix}\begin{bmatrix} v_1^e \\ \theta_1^e \\ v_2^e \\ \theta_2^e \end{bmatrix} = \begin{bmatrix} f_1^e \\ m_1^e \\ v_2^e \\ m_2^e \end{bmatrix}. \tag{14.33}$$

Taking the eigenvalues of this matrix yields three zeros and one nonzero eigenvalue, which confirms that its rank is one. For $\xi_H = 0$, which means that the hinge is at the element midpoint, one gets the stiffness equations (13.14) that appear in IFEM Chapter 13 [247].

The foregoing calculations were carried out symbolically with the *Mathematica* script listed in Figure 14.9. This script makes use of the element derivation utility modules listed in Figure 14.6. The last four lines of the script test the $\xi_H = 0$ case against the TPE-based element derived in Chapter 13 of IFEM, as previously noted.

Example 14.2. The element stiffness equations (14.33) are applied to a simple design optimization problem that can be entirely done by hand in symbolic form. This is pictured in Figure 14.10(a). A fixed-fixed prismatic beam AB of total span L is loaded by a point force P (positive upward) at its center C. Two hinges are symmetrically placed at distances $\frac{1}{2}\alpha L$ from the end supports. Because of the symmetry about C, it is possible to model the problem with just *one* hinged beam element, say the left half-span, as illustrated in Figure 14.10(b). The displacement support conditions are: $v_1 = \theta_1 = \theta_2 = 0$. This leaves only one unknown DOF: the midspan deflection $v_2 = v_C$ under P. Question: for which hinge configuration, defined by α as design variable, is v_C minimized in magnitude?



FIGURE 14.10. Application of the ESH-formulated hinged plane beam element to a simple design optimization problem.

Mathematically this is a constrained optimization problem:

$$\text{find} \quad \min_{\alpha} |v_C(\alpha)| \qquad \text{for } \alpha \in [0,1]. \tag{14.34}$$

Note that for the beam element depicted in Figure 14.10(b), $\xi_H = 2\alpha - 1$ and $\alpha = (1+\xi_H)/2$.

Does such a minimum exist? For both $\alpha = 0$ (a simply supported beam) and $\alpha = 1$ (a doubly hinged cantilever), Mechanics of Material textbooks give $v_C = -PL^3/(48EI_{zz})$, which provide a double check on your computations by setting $\xi_H = \pm 1$. For α in between, $|v_C|$ gets smaller, so there is in fact an optimal value, which is readily computable by hand.

Why is this configuration of interest to engineers? Minimizing v_C for fixed P means that the external energy $V = P\,v_C$ is minimized. Hence the effective rigidity under that loading is *maximized*. In structural optimization this is called a *maximum stiffness design*, which is of interest for certain classes of structures.

The stiffness equations before applying BC are (14.34), in which we replace $\xi_H \to 2\alpha - 1$, $L^e \to L/2$, $f_2^e \to -P/2$ and element superscript e may be removed. Applying the BCs $v_1 = \theta_1 = \theta_2 = 0$ and solving for v_2 we get

$$v_2 = v_C = \frac{\frac{1}{2}P}{12EI_{zz}/((L^e)^3(1+3\xi_H^2))} = \frac{P\,(L^e)^3(1+3\xi_H^2)}{24\,EI_{zz}} = \frac{P\,L^3(1+3\xi_H^2)}{192EI_{zz}}. \tag{14.35}$$

Evidently $|v_C|$ is minimized for $\xi_H = 0$, or $\alpha = 1/2$. That is, the hinges should be placed at the quarterspans. If this is done, the deflection is $PL^3/(192EI_{zz})$ as compared to $PL^3/(48EI_{zz})$ when $\xi_H = \pm 1$. That is a gain of 4 in deflection stiffness under the center point load.

Example 14.3. For the previous example configuration, defined in Figure 14.10, find the internal deflection $v(x)$ and cross section rotation $\theta(x) = v'(x)$ as function of the hinge position, and plot those functions, as well as the moment $M(x)$, for selective values of α.

Those functions can be found by integrating $M(x)/(EI_{zz})$ twice in x, since for a BE beam model, $v''(x) = \theta'(x) = M(x)/(EI_{zz})$. Care must be exercised, however, in applying kinematic BCs because the rotation $\theta(x) = v'(x)$ may be discontinuous at the hinge location, whereas both $M(x)$ and $v(x)$ are continuous there.

```
ClearAll[EI,Le,P,ξH,ξ,x,xx,Mref,A1,A2,B1,B2,c];
A1=0; A2=-P*Le^2*ξH/(4*EI); B1=0; B2=Le^3*P*ξH*(1+ξH)/(8*EI);
vC=(P/2)*Le^3*(1+3*ξH^2)/(12*EI);
G=Mref*{{2/Le,1+ξH,-2/Le,1-ξH}};
F={{(Le*Mref^2*(1+3*ξH^2))/(3*EI)}};
ue={0,0,vC,0}; {βe}=Simplify[Inverse[F].G.ue];
Me=βe*Mref*(ξ-ξH); Print["βe=",βe,"  Me=",Me]; Mxe=Me/.ξ->(2*x-Le)/Le;
θxeL=Simplify[Integrate[(Mxe/EI)/.x->xx,{xx,0,x}]+A1];
θxeR=Simplify[Integrate[(Mxe/EI)/.x->xx,{xx,0,x}]+A2];
Print["θxeL=",θxeL," θxeL at x=0: ",Simplify[θxeL/.x->0]];
Print["θxeR=",θxeR," θxeR at x=Le: ",Simplify[θxeR/.x->Le]];
vxeL=Simplify[Integrate[Simplify[θxeL/.x->xx],{xx,0,x}]+B1];
vxeR=Simplify[Integrate[Simplify[θxeR/.x->xx],{xx,0,x}]+B2];
Print["vxeL=",vxeL," vxeL at x=0:", Simplify[vxeL/.x->0]];
Print["vxeR=",vxeR," vxeR at x=Le:",Simplify[vxeR/.x->Le]];
dvxH=Simplify[(vxeR-vxeL)/.x->((1+ξH)/2)*Le];
dθxH=Simplify[(θxeR-θxeL)/.x->((1+ξH)/2)*Le];
{θeL,θeR}=Simplify[{θxeL,θxeR}/.x->((1+ξ)/2)*Le];
{veL,veR}=Simplify[{vxeL,vxeR}/.x->((1+ξ)/2)*Le];
Print["θeL=",θeL," θeR=",θeR]; Print["veL=",veL," veR=",veR];
Print["θeLA=",Simplify[θeL/.ξ->-1]," θeRC=",Simplify[θeR/.ξ->1]];
Print["veLA=",Simplify[veL/.ξ->-1]," veRC=",Simplify[veR/.ξ->1]];
style={{AbsoluteThickness[2],RGBColor[1,0,0]},
       {AbsoluteThickness[2],RGBColor[0,1,1/3]},
       {AbsoluteThickness[2],RGBColor[0,0,0]},
       {AbsoluteThickness[2],RGBColor[2/3,0,1]},
       {AbsoluteThickness[2],RGBColor[0,0,1]}};
colors=" R:α=0, G:α=1/4, Bk:α=1/2, V:α=3/4, Bl:α=1";
d0={Le->1,EI->1,P->1,ξH->-1};
d1={Le->1,EI->1,P->1,ξH->-1/2}; d2={Le->1,EI->1,P->1,ξH->0};
d3={Le->1,EI->1,P->1,ξH-> 1/2}; d4={Le->1,EI->1,P->1,ξH->1};
{veL0,veL1,veL2,veL3,veL4}={veL/.d0,veL/.d1,veL/.d2,veL/.d3,veL/.d4};
{veR0,veR1,veR2,veR3,veR4}={veR/.d0,veR/.d1,veR/.d2,veR/.d3,veR/.d4};
ve0=If[ξ<=-1,veL0,veR0]; ve1=If[ξ<=-1/2,veL1,veR1];
ve2=If[ξ<=0, veL2,veR2]; ve3=If[ξ<= 1/2,veL3,veR3];
ve4=If[ξ<=1,veL4,veR4];
Plot [{ve0,ve1,ve2,ve3,ve4},{ξ,-1,1}, PlotStyle->style,
      PlotLabel->"Deflection "<>colors, Frame->True,
      PlotRange->{0,0.2}, ImageSize->350];
{θeL0,θeL1,θeL2,θeL3,θeL4}={θeL/.d0,θeL/.d1,θeL/.d2,θeL/.d3,θeL/.d4};
{θeR0,θeR1,θeR2,θeR3,θeR4}={θeR/.d0,θeR/.d1,θeR/.d2,θeR/.d3,θeR/.d4};
θe0=If[ξ<=-1,θeL0,θeR0]; θe1=If[ξ<=-1/2,θeL1,θeR1];
θe2=If[ξ<=0, θeL2,θeR2]; θe3=If[ξ<= 1/2,θeL3,θeR3];
θe4=If[ξ<=1, θeL4,θeR4];
Plot [{θe0,θe1,θe2,θe3,θe4},{ξ,-1,1}, PlotStyle->style,
      PlotLabel->"Rotation "<>colors, Frame->True,
      PlotRange->{0,0.3}, ImageSize->350];
{Me0,Me1,Me2,Me3,Me4}={Me/.d0,Me/.d1,Me/.d2,Me/.d3,Me/.d4};
Plot [{Me0,Me1,Me2,Me3,Me4},{ξ,-1,1}, PlotStyle->style,
      PlotLabel->"Moment "<>colors, Frame->True,
      PlotRange->{-0.5,0.5}, ImageSize->350];
```

FIGURE 14.11. *Mathemaatica* script for solving Example 14.3 and generate the requested plots.

The solution is found and plotted through the *Mathematica* script of Figure 14.11. Here are some selected transcribed results: Recover bending moment from the displacement solution:

$$\beta_1^e = \mathbf{F}^{-1}\mathbf{G}\,\mathbf{u}^e = -\frac{P\,L^e}{4M_{ref}}, \quad M^e(\xi) = -\tfrac{1}{4}P\,L^e\,(\xi-\xi_H), \quad M^e(x) = -\tfrac{1}{4}P\,L^e\left(\frac{2x}{L^e}-1-\xi_H\right). \tag{14.36}$$

The latter is integrated twice with respect to x, and kinematic BC are applied (with care) to determine the



FIGURE 14.12. Moment, rotation and deflection plots produced by the script of Figure 14.11, taking $L^e = 1$, $EI_{zz} = 1$ and $P = 1$. Curve colors red, green, black, violet and blue pertain to $\alpha = 0$, $\frac{1}{4}$, $\frac{1}{2}$, $\frac{3}{4}$ and 1, respectively.

integration constants. Rotation functions in x:

$$\theta^e_L(x) = \frac{P\,x\,(L^e - x + L^e\,\xi_H)}{4E\,I_{zz}} \quad \text{for } x \in [0, x_H], \qquad \theta^e_R(x) = -\frac{P\,(L^e - x)(L^e\xi_H - x)}{4E\,I_{zz}} \quad \text{for } x \in [x_H, L^e]. \tag{14.37}$$

in which $x_H = \frac{1}{2}\alpha L = \alpha L^e = \frac{1}{4}(1 + 2\xi_H)L = \frac{1}{2}(1 + 2\xi_H)L$ is the x coordinate of the hinge. These satisfy two kinematic BC: $\theta^e_L(0) = 0$ and $\theta^e_R(L^e) = 0$. At $x = x_H$ there is generally a rotation discontinuity: $\theta^e_L(x_H) \neq \theta^e_R(x_H)$ except by chance. Deflection functions in x:

$$\begin{aligned} v^e_L(x) &= \frac{P\,x^2(3L^e(1 + \xi_H) - 2x)}{4E\,I_{zz}} \quad \text{for } x \in [0, x_H], \\ v^e_R(x) &= -\frac{P\left(3L^e\,x^2(1 + \xi_H) + 3(L^e)^3\,\xi_H\,(1 + \xi_H) - 6(L^e)^2\,x\,\xi_H - 2x^3\right)}{4E\,I_{zz}} \quad \text{for } x \in [x_H, L^e]. \end{aligned} \tag{14.38}$$

These satisfy two kinematic BC: $v^e_R(L^e) = 0$ and $v^e_L(x_H) = v^e_R(x_H)$. The latter BC says that $v^e(x)$ must be continuous at $x = x_H$. Rotation functions in $\xi = 2x/L^e - 1$:

$$\begin{aligned} \theta^e_L(\xi) &= \frac{P\,(L^e)^2\,(1 + \xi)(1 - \xi + 2\xi_H)}{16\,E\,I_{zz}} \quad \text{for } \xi \in [-1, \xi_H], \\ \theta^e_R(\xi) &= \frac{P\,(L^e)^2\,(1 - \xi)(1 + \xi - 2\xi_H)}{16\,E\,I_{zz}} \quad \text{for } \xi \in [\xi_H, 1]. \end{aligned} \tag{14.39}$$

Deflection functions in $\xi = 2x/L^e - 1$:

$$\begin{aligned} v^e_L(\xi) &= \frac{P\,(L^e)^3\,(1 + \xi)^2(2 - \xi + 3\xi_H)}{96\,E\,I_{zz}} \quad \text{for } \xi \in [-1, \xi_H], \\ v^e_R(\xi) &= \frac{P\,(L^e)^3\left(2 - \xi^3 + 3\xi_H + 3\xi^2\,\xi_H - 12\xi_H^2 + \xi\,(3 - 6\xi_H)\right)}{96\,E\,I_{zz}} \quad \text{for } \xi \in [\xi_H, 1]. \end{aligned} \tag{14.40}$$

Plots of $M^e(\xi)$, $\theta^e(\xi)$, and $v^e(\xi)$ over $\xi \in [-1, 1]$ are displayed in Figure 14.12 for hinge positions at $\alpha = 0$, $\frac{1}{4}$, $\frac{1}{2}$, $\frac{3}{4}$, and 1, which correspond to $\xi_H = -1$, $-\frac{1}{2}$, 0, $\frac{1}{2}$, and 1, respectively.

The deflection plots visually corroborate the main result of Example 14.2: the central deflection is minimized if $\alpha = \frac{1}{2}$ or $\xi_H = 0$. Note that for that particular configuration there is *no* rotational discontinuity at the hinge location: $\theta^e(\xi)$ is then a smooth continuous function. This happens to be the reason why the pure displacement (TPE) formulation of §13.2.5 of IFEM [247] happens (fortuitously) to work. Should the hinge not be at the center of the element, the TPE formulation given therein would give the wrong results.

Note also (by inspection) that the maximum absolute moment $\max_{\xi\in[-1,1]} |M(\xi, \alpha)|$ is minimized for $\alpha = \frac{1}{2}$.